
Change 1 
	Function name
	f_TC_19_4_5_IMS2   

	Reason for change
	After failure of the emergency registration, the UE may continue to use the TCP connection to the unprotected port, that it had opened for the initial REGISTER request, for executing the rest of the test procedure. Therefore, the SS should not close the unprotected port after the failed emergency registration but rather leave the UE the freedom to close the TCP connection or continue using the current connection.  

	Summary of change
	Added a new flag as input parameter to f_GlobalRelease in order to not close unprotected TCP ports when the flag is set to true.

	TTCN module
	IMS_IRAT/19/IMS_IRAT_19_IMS.ttcn 

	MCC160 Comment
	


Before change

  function f_TC_19_4_5_IMS2() runs on IMS_PTC

  { /* Emergency call without emergency registration / UE credentials are not accepted */

    var REGISTER_Request v_RegisterRequest;

    var INVITE_Request v_InviteRequest;

    var boolean v_IsUnprotected := true;

    f_IMS_PTC_Init(PDN_2);

    // Test Case Body

    f_IMS_TestBody_Set(true);

    //@siclog "Steps 2-17" siclog@

    // Emergency registration: 34.229-1 Annex C.20

    v_RegisterRequest := f_IMS_EmergencyCallRegistration_403Forbidden();

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 2 to 17");

    f_IMS_GlobalRelease();   // @sic R5s141339 change 2 - additional change sic@ */

    //@siclog "Steps 18-22" siclog@

    // Emergency call setup: 34.229-1 Annex C.22

    v_IsUnprotected := true; // Continue emergency call on non protected ports (but on second PDN)

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22(A_2_1_A6);  // @sic R5-131470 sic@ @sic R5s150846 sic@

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 18 to 22");

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    // Wait for Call Release Trigger

    f_IMS_IPCAN_WaitForTrigger(IPCAN);  // @sic R5s140786 sic@

    //@siclog "Steps 23-24" siclog@

    //Release Emergency Call MO

    f_IMS_EmergencyCallRelease(v_InviteRequest);  // @sic R5-131470 sic@

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 23 to 24");

    // Inform EUTRA: IMS Call released.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_TestBody_Set(false);   //IMS deregistration handled in default  @sic R5s160125 sic@

  }      
After change

  function f_TC_19_4_5_IMS2() runs on IMS_PTC

  { /* Emergency call without emergency registration / UE credentials are not accepted */

    var REGISTER_Request v_RegisterRequest;

    var INVITE_Request v_InviteRequest;

    var boolean v_IsUnprotected := true;

    f_IMS_PTC_Init(PDN_2);

    // Test Case Body

    f_IMS_TestBody_Set(true);

    //@siclog "Steps 2-17" siclog@

    // Emergency registration: 34.229-1 Annex C.20

    v_RegisterRequest := f_IMS_EmergencyCallRegistration_403Forbidden();

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 2 to 17");

    f_IMS_GlobalRelease(true);   // @sic R5s141339 change 2 - additional change sic@ */
    //@siclog "Steps 18-22" siclog@

    // Emergency call setup: 34.229-1 Annex C.22

    v_IsUnprotected := true; // Continue emergency call on non protected ports (but on second PDN)

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22(A_2_1_A6);  // @sic R5-131470 sic@ @sic R5s150846 sic@

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 18 to 22");

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    // Wait for Call Release Trigger

    f_IMS_IPCAN_WaitForTrigger(IPCAN);  // @sic R5s140786 sic@

    //@siclog "Steps 23-24" siclog@

    //Release Emergency Call MO

    f_IMS_EmergencyCallRelease(v_InviteRequest);  // @sic R5-131470 sic@

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 23 to 24");

    // Inform EUTRA: IMS Call released.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_TestBody_Set(false);   //IMS deregistration handled in default  @sic R5s160125 sic@

  }     

Change 2 
	Function name
	f_IMS_GlobalRelease   

	Reason for change
	See Change 1  

	Summary of change
	Added a new flag as input parameter to f_GlobalRelease. When flag is set to true, new function f_IMS_RegInfo_WithoutUnprotectedclose_Release is called in order to not close the TCP connection.   

	TTCN module
	Common/IMS/IMS_Procedures_Common.ttcn 

	MCC160 Comment
	


Before change

  function f_IMS_GlobalRelease() runs on IMS_PTC

  {

    f_IMS_RegisterInfo_Release();  /* release registration information at IPCAN */

    f_IMS_PTC_Reset();             /* reset ImsInfo and release security

                                      @sic R5s150031: common function f_IMS_PTC_Reset sic@ */

  }      

After change

  function f_IMS_GlobalRelease(boolean p_nonUnprotected:= false) runs on IMS_PTC

  {

    if (p_nonUnprotected) {

    
f_IMS_RegInfo_WithoutUnprotectedclose_Release();

    } else {
  
  f_IMS_RegisterInfo_Release();  /* release registration information at IPCAN */

    }
    f_IMS_PTC_Reset();             /* reset ImsInfo and release security

                                      @sic R5s150031: common function f_IMS_PTC_Reset sic@ */

  }     
Change 3 
	Function name
	New function: f_IMS_RegInfo_WithoutUnprotectedclose_Release   

	Reason for change
	Implement the functionality to not close TCP connection  

	Summary of change
	Made new function correspond to existing f_IMS_RegisterInfo_Release with a difference in the first argument to embedded call to fl_IMS_ConfigureIP  

	TTCN module
	Common/IMS/IMS_IP_Config.ttcn 

	MCC160 Comment
	


Before change

n/a      
After change

  /*

   * @desc      trigger IP PTC to release registration info related to this IMS PTC without closing the unprotected ports

   * 

   */

  function f_IMS_RegInfo_WithoutUnprotectedclose_Release() runs on IMS_PTC

  {

    fl_IMS_ConfigureIP(cas_IMS_RegInfoRelNonUnprotected_REQ,

                       car_IMS_RegInfoRelease_CNF);  /* @sic R5s140780 - MCC160 Implementation sic@ */

  }     
Change 4 
	Function name
	New template:    cas_IMS_RegInfoRelNonUnprotected_REQ

	Reason for change
	Indicate that connection should not be closed  

	Summary of change
	Defined new template that is a mirror of existing cas_IMS_ReginInfoRelease_REQ, except that RegInfoRelease is not set to true but to false  

	TTCN module
	Common/IMS/IMS_IP_Config.ttcn 

	MCC160 Comment
	


Before change

n/a      
After change

template (value) IMS_CONFIG_REQ cas_IMS_RegInfoRelNonUnprotected_REQ := {RegInfoRelease := false};     

Change 5 
	Function name
	car_IMS_RegInfoRelease_REQ   

	Reason for change
	Receive template for indication of connection release should not be hardwired to true but be flexible to allow also the case of not closing the connection  

	Summary of change
	Replaced hardwired value by any value being allowed  

	TTCN module
	Common/IP_PTC/IP_PTC_IMS_Templates.ttcn 

	MCC160 Comment
	


Before change

template (present) IMS_CONFIG_REQ car_IMS_RegInfoRelease_REQ := {RegInfoRelease := true};   
After change

template (present) IMS_CONFIG_REQ car_IMS_RegInfoRelease_REQ := {RegInfoRelease := ?};     
Change 6 
	Function name
	a_IP_IMS_Config

	Reason for change
	Close TCP connection only if configured that way in IP_PTC  

	Summary of change
	Added a check of indication boolean to see if connection should be closed or not.  

	TTCN module
	Common/IP_PTC/IP_PTC/IP_PTC_IMS_Handler.ttcn 

	MCC160 Comment
	


Before change

  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue (v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */

          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);

          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ  /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveAllProtectedSecurityContexts(p_ImsServer, v_WaitForCNF);            /* @sic R5s141154, R5s151044 sic@ */

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, v_WaitForCNF);          /* @sic R5s151044 sic@ */
        p_ImsServer.RegistrationAddress := omit;

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }      

After change

  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue (v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */

          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);

          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ  /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveAllProtectedSecurityContexts(p_ImsServer, v_WaitForCNF); /* @sic R5s141154, R5s151044 sic@ */


     if(v_IMS_CONFIG_REQ.RegInfoRelease == true){         

             fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, v_WaitForCNF);          /* @sic R5s151044 sic@ */

        }
        p_ImsServer.RegistrationAddress := omit;

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }     

