
1. Change – Removal of fl_ImsServer_UeIsRegistered
	Reason for change
	fl_ImsServer_UeIsRegistered may cause confusion as at the IP PTC it is in general not known whether or not the UE is IMS registered.

But it is known after a first IMS register request whether the UE uses IPv4 or IPv6 for IMS signalling

NOTE: It is assumed that the UE does not change IPv4/IPv6 for IMS during a test case

	Summary of change
	· fl_ImsServer_UeIsRegistered removed;

· fl_IMS_Server_GetServerAddrRX, fl_IP_IMS_Server_SendDL:
call to fl_ImsServer_UeIsRegistered replaced by "ispresent(p_ImsServer.RegistrationAddress)" 

· a_IP_IMS_Server_ReceiveTCP, a_IP_IMS_Server_ReceiveUDP:
no need not to forward IP addresses to IMS in the RoutingInfo of the UL ASP

	TTCN module
	IP_PTC_IMS_Handler


	




	  function fl_IMS_Server_GetServerAddrRX(IMS_Server_Type p_ImsServer) return template (present) IP_AddrInfo_Type

  {

    var template (present) IP_AddrInfo_Type v_ServerAddr := ?;

    if (f_ImsServer_IsStarted(p_ImsServer)) {


      if (ispresent(p_ImsServer.RegistrationAddress)) {
        v_ServerAddr := p_ImsServer.RegistrationAddress.NW;

      } else {

        v_ServerAddr := (p_ImsServer.IPv4Addr, p_ImsServer.IPv6Addr);

      }

    }

    return v_ServerAddr;

  }

	  altstep a_IP_IMS_Server_ReceiveTCP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     IMS_SecurityContextEnum_Type p_SecurityContextEnum) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var template (omit) IMS_SecurityContext_Type v_SecurityContext := fl_IMS_Server_GetSecurityContext(p_ImsServer, p_SecurityContextEnum);

    var boolean v_ClientIsConnected := isvalue(v_SecurityContext.TcpClient);

    var boolean v_ServerIsConnected := isvalue(v_SecurityContext.TcpServer);

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var Datagram_Content_Type v_Data;

    [v_ClientIsConnected] IP_SOCK_DATA.receive(cr_TCP_DATA_IND(valueof(v_SecurityContext.TcpClient)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.TCP.Recv;

        v_RoutingInfo := cs_IMS_RoutingInfo(tcp, p_SecurityContextEnum);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, response, v_RoutingInfo, v_Data);

      }

    [v_ServerIsConnected] IP_SOCK_DATA.receive(cr_TCP_DATA_IND(valueof(v_SecurityContext.TcpServer)))

      -> value v_IP_SOCKET_IND

      {



          v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    // may be IPv4 or IPv6

          v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   // may be IPv4 or IPv6


        v_Data := v_IP_SOCKET_IND.DATA.Ind.TCP.Recv;

        v_RoutingInfo := cs_IMS_RoutingInfo(tcp, p_SecurityContextEnum, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, request, v_RoutingInfo, v_Data);  /* @sic R5s141322 change 1 sic@ */

      }

  }

	  altstep a_IP_IMS_Server_ReceiveUDP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     IMS_SecurityContextEnum_Type p_SecurityContextEnum) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var template (present) IP_AddrInfo_Type v_ServerAddr := fl_IMS_Server_GetServerAddrRX(p_ImsServer);

    var template (omit) IMS_SecurityContext_Type v_SecurityContext := fl_IMS_Server_GetSecurityContext(p_ImsServer, p_SecurityContextEnum);

    var template (present) IMS_LocalRemotePort_Type v_LocalRemotePort := fl_IP_IMS_GetPortsRX(v_SecurityContext);

    var boolean v_UdpServerIsStarted := isvalue(v_SecurityContext);

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var Datagram_Content_Type v_Data;

    var IMS_RequestResponse_Type v_RequestResponse;

    // receive UDP data

    [v_UdpServerIsStarted] IP_SOCK_DATA.receive(cr_UDP_DATA_IND(cr_IP_Socket(v_ServerAddr, v_LocalRemotePort.Local),

                                                                cr_IP_Socket(?, v_LocalRemotePort.Remote)))  /* remote port is ? for unprotected connection;

                                                                                                                for protected connection the remote port is needed to distinguish the context */

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;

        v_RequestResponse := fl_IP_IMS_UL_ResponseOrRequest(v_Data); /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                      * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                      * cannot be distinguished by the port on which they have been received */
        if (v_RequestResponse == request) {

          v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;

          v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;

        }











        v_RoutingInfo := cs_IMS_RoutingInfo(udp, p_SecurityContextEnum, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_RequestResponse, v_RoutingInfo, v_Data);

      }

  }

	  function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,

                                   IMS_RoutingInfo_Type p_RoutingInfo,

                                   IMS_RequestResponse_Type p_RequestResponse,

                                   bitstring p_Data) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_SecurityContextEnum_Type v_SecurityContextEnum;

    var IMS_SecurityContext_Type v_SecurityContext;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var IP_Connection_Type v_NewTcpConnectionByRef;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_IsUnprotected;

    var IP_DrbInfo_Type v_DrbInfo;

    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_SecurityContextEnum := p_RoutingInfo.Security;

    } else {

      if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {         /* the latest security context is preferred */

        v_SecurityContextEnum := protectedContext2;

      } else if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

        v_SecurityContextEnum := protectedContext1;

      } else {

        v_SecurityContextEnum := unprotected;

      }

    }

    if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, v_SecurityContextEnum)) {

      FatalError(__FILE__, __LINE__, "IMS security not established");

    }

    v_SecurityContext := valueof(fl_IMS_Server_GetSecurityContext(p_ImsServer, v_SecurityContextEnum));

    v_IsUnprotected := (v_SecurityContextEnum == unprotected);

    v_LocalRemotePort := fl_IP_IMS_GetPortsTX(v_SecurityContext, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;


    } else if (ispresent(p_ImsServer.RegistrationAddress)) {
      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK_DATA.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (p_RequestResponse) {

          case (response) {

            v_TCP_Connection := v_SecurityContext.TcpServer;

          }

          case (request) {

            v_TCP_Connection := v_SecurityContext.TcpClient;

            if (not isvalue(v_TCP_Connection)) {

              if (v_IsUnprotected) {

                v_LocalPort := omit;  // in case of an unprotected TCP connection to UE there is no need to specify any specific local port but we can use an ephemeral port instead

              }

              v_NewTcpConnectionByRef := f_TcpClient_ConnectReq(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), ims);

              v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(p_ImsServer.PdnIndex);   /* @sic R5s150648 Change 3 sic@ */

              alt {

                [] a_IP_IMS_TCP_ConnectionControl(p_ImsServer) {  /* @sic R5s140713 - MCC160 implementation sic@ */

                  repeat;

                }

                [] a_TcpClient_ConnectCnf(v_NewTcpConnectionByRef, v_DrbInfo) {

                  v_SecurityContext.TcpClient := v_NewTcpConnectionByRef;

                  fl_IMS_Server_SetSecurityContext(p_ImsServer, v_SecurityContextEnum, v_SecurityContext); // @sic R5s141322 change 2 sic@

                  v_TCP_Connection := v_NewTcpConnectionByRef;

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          /* @sic R5s150270 change 2: just skip the data sic@

             there can be a race condion when the UE does not wait for any 200 OK but immediately closes the TCP connection with the server

             => when TTCN wants to send the 200 OK the connection is gone already */

        } else {

          IP_SOCK_DATA.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }


2. Change – Simplified implementation for release of security context(s)
	Reason for change
	fl_IMS_Server_RemoveSecurityContext can easily be enhanced to remove either the oldest security context or all of them; this simplifies the implementation and improves readability

	Summary of change
	· fl_IMS_Server_RemoveSecurityContext enhanced to deal with one or all contexts
· fl_IMS_Server_RemoveAllProtectedSecurityContexts removed
· "p_ImsServer.RegistrationAddress := omit" removed from the car_IMS_RegInfoRelease_REQ branch in a_IP_IMS_Config (see change 1)

	TTCN module
	IP_PTC_IMS_Handler


	  type enumerated IMS_RemoveSecurityContext_Type {allProtected, oldestOnly};

	  function fl_IMS_Server_RemoveSecurityContext(inout IMS_Server_Type p_ImsServer, 
                                               IMS_RemoveSecurityContext_Type p_AllOrOldestOnly := oldestOnly,
                                               boolean p_WaitForCNF := true) runs on IP_PTC

  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    /* @sic R5-145732: support of 2nd security context sic@ */

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    while (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

      v_SecurityInfo := p_ImsServer.ProtectedContext1.SecurityInfo.Protected;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      // Close TCP connection

      fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1, p_WaitForCNF);

      if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {   // if there is no other security context ...

        // ... stop TCP and UDP server

        f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

        f_TcpServer_Stop(v_NW_Address, v_LocalPortServer, p_WaitForCNF);

        p_ImsServer.ProtectedContext1 := omit;

      }

      else {

        p_ImsServer.ProtectedContext1 := p_ImsServer.ProtectedContext2;

        p_ImsServer.ProtectedContext2 := omit;

      }

      // Release security association

      fl_IPsec_Release(v_NW_Address, v_UE_Address, v_SecurityInfo); 
      if (p_AllOrOldestOnly == oldestOnly) {

        break;

      }
    }

  }

	









	  function f_IP_IMS_Stop(IMS_Server_Type p_ImsServer) runs on IP_PTC return IMS_Server_Type

  {

    var IMS_Server_Type v_ImsServer := p_ImsServer;

    if (f_ImsServer_IsStarted(v_ImsServer)) {


      fl_IMS_Server_RemoveSecurityContext(p_ImsServer, allProtected);
      f_UdpServer_Stop(v_ImsServer.IPv4Addr, tsc_IMS_PortNumber_5060);

      f_UdpServer_Stop(v_ImsServer.IPv6Addr, tsc_IMS_PortNumber_5060);

      fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

      f_TcpServer_Stop(v_ImsServer.IPv4Addr, tsc_IMS_PortNumber_5060);

      f_TcpServer_Stop(v_ImsServer.IPv6Addr, tsc_IMS_PortNumber_5060);

      v_ImsServer := valueof(cs_IMS_Server_Init);

    }

    return v_ImsServer;

  }

	  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var template (value) IMS_CONFIG_CNF cas_IMS_CONFIG_CNF;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue(v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */

          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);

          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      } 
    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, oldestOnly, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ  /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false; 
        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, allProtected, v_WaitForCNF);
           /* @sic R5s141154, R5s151044 sic@ */

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, v_WaitForCNF);          /* @sic R5s151044 sic@ */


        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }


